

Ecco il listato dell'avventura testuale [Lelé e la piccola Uku](#).

Si riferisce al corso realizzato dagli E-Papers Adventures sulla creazione di un'avventura testuale per il Vic-20, scritta sul Vic20 in live.

Sono ben 7+1(bonus) video su come si crea una bella avventura, strutturata su una mappa, con *quest* da risolvere e pericoli da superare.

Per chiunque fosse interessato, vi invito ad andare a vedere i loro video ed iscrivermi al loro canale YouTube per seguire tutti i loro lavori. Nel frattempo, vi spiego in anteprima la storia (spero di non sbagliare):

*Abiti nel villaggio dei capelli gialli, e devi trovare lo sciamano per farti dare la medicina, necessaria a salvare tuo nonno, gravemente malato.*

Di seguito riporto il loro programma, scritto durante il corso, ma con le ultime modifiche per evitare il fatidico OUT OF MEMORY ERROR. Il corso è piacevole e molto interessante, se avete voglia di imparare qualche cosa di nuovo, seguiteli e vedrete che si scoprono tante info interessanti.

Vediamo ora il programma diviso in parti:

```
100 PRINT"███LELE' E LA PICCOLA UKU█":P=1:SE=1:TN=4:RA=6:FI=2
110 PRINT"E-PAPER ADVENTURES '21█"
111 PRINT"TI CHIAMI UKU E VIVI NEL VILLAGGIO DEI CAP-PELLI GIALLI"
112 PRINT"TUO NONNO EKO E' MOLTOMALATO, DEVI TROVARE LO SCIAMANO"
114 PRINT"█USA VERBO+SOSTANTIVO E N,S,E,O,I,G,Q,█":GOTO200
150 N=0:S=0:E=0:O=0:PRINT"█"
200 IFP=1THENPRINT"SEI NEL TUO VILLAGGIO":N=2:E=3:CS=31
201 IFP=2THENPRINT"TI TROVI IN UN CAMPO FIORITO":S=1:CS=28
202 IFP=3THENPRINT"STAI CAMMINANDO IN UNA INTRICATA FORESTA":O=1:E=4:S=6:CS=29
203 IFP=4THENPRINT"SEI SULLA RIVA DI UN FIUME SINUOSO":O=3:CS=30
204 IFP=5THENGOTO5000
205 IFP=6THENPRINT"DI FRONTE A TE C'E' UNA CASCATA":N=3:CS=27
206 IFP=7THENPRINT"QUI, OLTRE IL FIUME, C'E' UNA BUIA GROTTA.":CS=24
230 POKE36879,CS:PRINT"QUI VEDI":X=P:GOSUB1000
231 IFP=3ANDSE=-1ANDQS=0THENGOSUB3000
400 INPUT"█E ORA";A$:MO=MO+1:PRINT
410 IFA$="N"ANDN>0THENP=N:GOTO150
411 IFA$="S"ANDS>0THENP=S:GOTO150
412 IFA$="E"ANDE>0THENP=E:GOTO150
413 IFA$="O"ANDO>0THENP=O:GOTO150
450 IFA$="I"THENX=-1:PRINT"HAI CON TE":GOSUB1000:GOTO400
490 IFQR<>1ANDA$="USA SECCHIO"ANDP=6ANDSE=-1THENGOTO3300
491 IFQR=1ANDA$="USA SECCHIO"ANDP=6ANDSE=-1THENGOTO3600
```

```

492 IFQR<>1ANDA$="PARLA RANA"ANDP=6THENGOTO3300
493 IFSC=PANDA$="PARLA SCIMMIA"THENPRINT"<<UAH UAH UAAAAHHH!>>":GOTO400
494 IFSC=PANDA$="UCCIDI SCIMMIA"THENPRINT"USA LA TESTA INVECE!":GOTO400
495 IF(SP=PORSP=-1)ANDA$="PARLA PESCIOLINO"THENPRINT"<<DOVE MI PORTI?>>":GOTO400
500 IFA$="PRENDI SECCHIO"ANDSE=PTHENSE=-1:GOTO2000
501 IFA$="LASCIA SECCHIO"ANDSE=-1THENSE=P:GOTO2000
502 IFA$="PRENDI SECCHIO"ANDSP=PTHENSP=-1:GOTO2000
503 IFA$="LASCIA SECCHIO"ANDSP=-1THENSP=P:GOTO2000
550 IFA$="NUOTA"ANDP=7THENP=4:GOTO150
551 IFA$="ENTRA"ANDP=7THENP=5:GOTO150
600 IFQS=1ANDA$="TOGLI CAPPELLO"ANDP=3THENGOTO3100
610 IFQT<>1ANDA$="NUOTA"ANDP=4THENPRINT"IL TONNO E' INSUPERA- BILE":GOTO400
611 IFQT=1ANDA$="NUOTA"ANDP=4THENP=7:GOTO150
620 IFQT<>1ANDA$="PARLA TONNO"ANDP=4THENGOTO3200
699 IFA$="PRENDI FIORE"AND(PF=0ORPR=0)ANDFI=PTHENPRINT"HAI UN BUON MOTIVO?":GOTO400
700 IFA$="PRENDI FIORE"ANDFI=PTHENFI=-1:GOTO2000
701 IFA$="DAI FIORE"ANDQR=0ANDFI=-1ANDP=6THENGOTO3500
702 IFA$="LASCIA FIORE"ANDFI=-1THENFI=P:GOTO2000
703 IFA$="PARLA FIORE"AND(FI=-1ORFI=P)THENGOTO3400
980 IFA$="Q"THENPRINT"ADDIO!":END
981 IFA$="G"THENGOTO150
990 IFA$="N"ORA$="S"ORA$="E"ORA$="O"THENGOTO999
991 IFLEN(A$)<3THENGOTO998
993 A$=LEFT$(A$,3):IFA$="USA"ORA$="UCC"ORA$="PRE"ORA$="LAS"ORA$="PAR"ORA$="TOG"THENGOTO999
994 IFA$="DAI"ORA$="RIE"ORA$="ENT"ORA$="NUO"THENGOTO999
998 PRINT"NON CAPISCO":GOTO400
999 PRINT"NON PUOI":GOTO400
1000 Y=0:IFSE=XTHENPRINT"UN SECCHIO":Y=1
1002 IFFI=XTHENPRINT"UN FIORE VIOLETTO":Y=1
1010 IFSC=XTHENPRINT"LA SCIMMIA CON UN SEC-CHIO IN TESTA":Y=1
1020 IFTN=XTHENPRINT"UN TONNO INSUPERABILE":Y=1
1030 IFRA=XTHENPRINT"UNA RANA DEPRESSA":Y=1
1031 IFSP=XTHENPRINT"UN SECCHIO CON UN PE- SCIOLINO":Y=1
1050 IFY=0THENPRINT"NULLA"
1099 RETURN
2000 PRINT"OK":GOTO400
3000 PRINT"LOH NO! UNA SCIMMIA TI RUBA IL SECCHIO E SE LO METTE IN TESTA COMEUN CAPPELLO"
3001 SE=0:QS=1:SC=3:RETURN
3100 PRINT"LA SCIMMIA TI IMITA, LASCIA CADERE IL SEC- CHIO E SE NE VA"

```

```

3101 SE=3:SC=0:QS=2:GOTO400

3200 PRINT"<<BLUB! QUANTI SASSO- LINI CI SONO IN QUESTOFIUME?>>"

3201 IFSP<>-1THENGOTO400

3202 PRINT"[U]IL PESCIOLINO RISPONDE<<FACILE,";STR$(INT(RND(1)*444)*76);"345!>>"

3203 PRINT"[U]IL TONNO ANNUISCE E SENE VA. RINGRAZI IL PE-SCIOLINO E LO LIBERI"

3204 SP=0:QT=1:TN=0:GOTO400

3300 PRINT"<<CRA CRA! FAMMI COM- PAGNIA!>>":PR=1:GOTO400

3400 PRINT"BLA ";:GETA$:IFA$=""THENGOTO3400

3403 PF=1:PRINT:GOTO400

3500 PRINT"BLA BLA CRA CRA ";:GETA$:IFA$=""THENGOTO3500

3504 PRINT:PRINT"[U]<<CRAAAZIE! ORA SONO UNA RANA FELICE!>> I DUE SE NE VANNO PER MANINA"

3505 QR=1:FI=0:RA=0:GOTO400

3600 PRINT"RACCOGLI L'ACQUA DELLACASCATA. SORPRESA! NELSECCHIO ORA C'E' UN PESCIOLINO"

3601 SE=0:SP=6:GOTO400

5000 PRINT"[U]":POKE36879,8

5004 INPUTA$:IFA$<>"LELE'"ANDA$<>"CHIAMA LELE'"THENGOTO5004

5010 FORY=0TO255:POKE36879,Y:NEXTY

5015 POKE36879,31:PRINT"[U][U]COMPLIMENTI!!![U]"

5019 PRINT"LELE' TI DONA LA MEDI-CINA[U]"

5021 PRINT"HAI RISOLTO IN";STR$(MO):PRINT"MOSSE"

```

## DESCRIZIONE DEL PROGRAMMA

- Dalla riga 100 : 114 = presentazione del gioco con intro, sempre alla 114 si salta alla 200.  
 Alla riga 100 si settano le posizioni dei vari oggetti con cui interagire.
- La riga 115 = azzerare le variabili di movimento Nord-Sud-Est-Ovest, poi cancellare lo schermo.
- Dalla riga 200 : 231 = vengono definite le stanze/luoghi e si settano le varie uscite disponibili, si definiscono i colori dello schermo in funzione della stanza, per mezzo della variabile CS.
- Dalla riga 400 : 999 = vengono definite tutte le varie azioni - creazione del parser dell'avventura.  
 Sono definite tutte le quest e le possibili azioni. Questo è il punto nevralgico dell'intera avventura. Bisogna settare tutte le clausole per fare in modo che gli eventi dell'avventura girino come si deve.
- Dalla riga 1000 : 1099 = è la parte di codice che gestisce l'inventario
- Dalla riga 3000 : 3601 = vengono gestite le varie quest dell'avventura, si settano le variabili del gioco in funzione delle varie azioni.
- Dalla riga 5000 : fine = si arriva alla fine dell'avventura.

Il programma è abbastanza semplice da capire, vengono usati pochi comandi in BASIC, però trovo molto utile seguire il corso per capire passo-passo la scrittura del codice, viene molto più semplice da capire.

Per completezza, vi riporto la versione che ho scritto seguendo il corso per intero, è molto simile a quello definitivo ed ottimizzato.

Allegherò tutto il materiale possibile:

- file ottimizzato
- file scritto seguendo il corso
- indirizzi delle 8 puntate del corso

Buono studio, e vi garantisco che realizzerò un'avventura testuale seguendo i loro insegnamenti.

Ecco il codice realizzato seguendo il corso:

```
5 PRINTCHR$(147)"■"  
10 P=1: REM POSIZIONE INIZIALE VILLAGGIO  
11 SE=1: REM SECCHIO SEMPRE NEL VILLAGGIO  
12 FI=2: REM FIORE NEL CAMPO FIORITO  
13 PA=0: REM PESCE ARCOBALENO  
15 TN=4: REM TONNO INSUPERABILE  
16 RA=6: REM RANA DEPRESSA  
17 SP=0: REM SECCHIO CON PESCE ARCOBALENO  
20 QS=0  
21 SC=0  
22 QT=0  
23 QR=0  
  
100 PRINT  
110 PRINT"LELE' E LA PICCOLA UKU"  
111 PRINT"BY E-PAPER ADVENTURES '21"  
150 N=0:S=0:E=0:O=0  
151 PRINT  
200 IFP=1THENPRINT"SEI NEL VILLAGGIO DEI CAPPELLI GIALLI":N=2:E=3:CS=31  
201 IFP=2THENPRINT"TI TROVI IN UN CAMPO FIORITO":S=1:CS=28  
202 IFP=3THENPRINT"STAI CAMMINANDO IN UNA INTRICATA FORESTA":O=1:E=4:S=6:CS=29  
203 IFP=4THENPRINT"SEI SULLA RIVA DI UN FIUME SINUOSO":O=3:CS=30  
204 IFP=5THENGOSUB5000:S=7  
205 IFP=6THENPRINT"DI FRONTE A TE C'E' UNA CASCATA":N=3:CS=27  
206 IFP=7THENPRINT"SEI DAVANTI ALLA GROTTA DELLO SCIAMANO":CS=24  
230 POKE36879,CS  
248 PRINT"QUI VEDI:"  
249 X=P  
250 GOSUB1000:REM CONTROLLO OGGETTI  
260 GOSUB2000
```

```

400 PRINT
401 INPUT "COSA FAI";A$
410 IFA$="N"ANDN>0THENP=N:GOTO150
412 IFA$="S"ANDS>0THENP=S:GOTO150
414 IFA$="E"ANDE>0THENP=E:GOTO150
415 IFA$="O"ANDO>0THENP=O:GOTO150
450 IFA$="I"THENX=-1:PRINT"HAI CON TE":GOSUB1000:GOTO400
500 IFA$="PRENDI SECCHIO"ANDSE=PTHENSE=-1:PRINT"OK":GOTO400
501 IFA$="LASCIA SECCHIO"ANDSE=-1THENSE=P:PRINT"OK":GOTO400
502 IFA$="PRENDI SECCHIO"ANDSP=PTHENSP=-1:PRINT"OK":GOTO400
503 IFA$="LASCIA SECCHIO"ANDSP=-1THENSP=P:PRINT"OK":GOTO400
550 IFA$="NUOTA"ANDP=7THENP=4:GOTO150
551 IFA$="ENTRA"ANDP=7THENP=5:GOTO150
600 IF(QS=1)AND(A$="LASCIA CAPPELLO"ORA$="TOGLI CAPPELLO")THENGOSUB3100:GOTO400
610 IFQT<>1ANDA$="NUOTA"ANDP=4THENPRINT"IL TONNO E' INSUPERABILE":GOTO400
611 IFQT=1ANDA$="NUOTA"ANDP=4THENP=7:GOTO150
620 IFQT<>1ANDA$="PARLA COL TONNO"ANDP=4THENGOSUB3200:GOTO400
630 IFQR<>2ANDA$="USA SECCHIO CON CASCATA"ANDP=6ANDSE=-1THENGOSUB3300:GOTO400
631 IFQR=1ANDA$="USA SECCHIO CON CASCATA"ANDP=6ANDSE=-1THENGOSUB3600:GOTO400
700 IFA$="PRENDI FIORE"ANDFI=PTHENFI=-1:PRINT"OK":GOTO400
701 IFA$="LASCIA FIORE"ANDFI=-1THENFI=P:PRINT"OK":GOTO400
702 IFA$="PARLA COL FIORE"AND(FI=-1ORFI=P)THENGOSUB3400:GOTO400
710 IFA$="USA FIORE CON RANA"ANDQR=0ANDFI=-1ANDP=6THENGOSUB3500:GOTO400
980 IFA$="Q"THENPRINT"ADDIO!":END
981 IFA$="G"THENGOTO150
990 IFA$="N"ORA$="S"ORA$="E"ORA$="O"THEN998
991 IFLEN(A$)<3THENPRINT"NON CAPISCO":GOTO400
992 A$=LEFT$(A$,3)
993 IFA$="USA"ORA$="ESA"ORA$="PRE"ORA$="LAS"ORA$="PAR"ORA$="TOG"THEN998
998 PRINT"NON PUOI"
999 GOTO400
1000 Y=0
1001 IFSE=XTHENPRINT"UN SECCHIO":Y=1
1002 IFFI=XTHENPRINT"UN FIORE CHIACCIERINO":Y=1
1003 IFPA=XTHENPRINT"UN PESCE ARCOBALENO":Y=1
1010 IFSC=XTHENPRINT"UNA SCIMMIA CON UN SECCHIO IN TESTA":Y=1
1020 IFTN=XTHENPRINT"UN TONNO INSUPERABILE":Y=1
1030 IFRA=XTHENPRINT"UNA RANA DEPRESSA":Y=1
1031 IFSP=XTHENPRINT"UN SECCHIO CON PESCE ARCOBALENO":Y=1
1050 IFY=0THENPRINT"NULLA DI INTERESSANTE"
1099 RETURN
2000 IFP=3ANDSE=-1ANDQS=0THENGOSUB3000
2999 RETURN
3000 PRINT"HO NO!! UNA SCIMMIA TI RUBA IL SECCHIO E SE LO METTE IN TESTA"
3001 SE=0:QS=1:SC=3:RETURN
3100 PRINT"LA SCIMMIA TI IMITA E LASCIA CADERE IL SECCHIO E SE NE VA."

```

```

3101 SE=3:SC=0:QS=2:RETURN
3200 PRINT"QUANTI SASSOLINI CI SONO NEL FIUME"
3201 IFSP<>-1THENRETURN
3202 PRINT"IL PESCE RISPONDE <<FACILE,";STR$(INT(RND(1)*444)*76);"34! >>"
3203 PRINT"IL TONNO ANNUISCE E SE NE VA. RINGRAZI IL PESCE E LO LIBERI."
3204 SP=0:QT=1:TN=0
3205 GETA$:IFA$=""THEN3205
3206 RETURN
3300 PRINT "<<NON METTERE IL SECCHIO LI', CRA CRA! FAMMI COMPAGNIA>>"
3301 RETURN
3400 PRINT"BLA BLA BLA BLA BLA"
3401 GETA$:IFA$=""THEN3400
3403 PRINT:RETURN
3500 PRINT"BLA BLA CRA CRA"
3501 GETA$:IFA$=""THEN3500
3505 PRINT"CRAAAAZIE!!! ORA SONO UNA RANA FELICE"
3506 QR=1:FI=0:RA=0
3510 RETURN
3600 PRINT"RACCOLGI L'ACQUA DELLA CASCATA. SORPRESA NEL SECCHIO C'E' UN PESCE"
3601 SE=0:SP=6:RETURN
5000 POKE36879,8:PRINTCHR$(147)
5003 INPUTA$
5004 IFA$<>"LELE'"THENGOTO5003
5010 FORT=1TO2
5011 FORJ=0TO255
5012 POKE36879,J
5013 NEXTJ,T
5015 POKE36879,31
5016 PRINTCHR$(147)"👁BRAVISSIMA!!! LO SCIAMANO TI DA LA MEDICINA."
5017 PRINT"PER IL NONNO MALATO."
5020 PRINT
5021 PRINT"CIAO, ALLA PROSSIMA!"
5022 END

```

## INDIRIZZI DEL CORSO

Video 1:

[https://www.youtube.com/watch?v=rLj4vcFTL5g&list=PLvL\\_GB22RnWJqRXbYcC5uIpHzupeqx6Zj&index=1](https://www.youtube.com/watch?v=rLj4vcFTL5g&list=PLvL_GB22RnWJqRXbYcC5uIpHzupeqx6Zj&index=1)

Video 2:

[https://www.youtube.com/watch?v=\\_2y2rp7TP7k&list=PLvL\\_GB22RnWJqRXbYcC5uIpHzupeqx6Zj&index=2](https://www.youtube.com/watch?v=_2y2rp7TP7k&list=PLvL_GB22RnWJqRXbYcC5uIpHzupeqx6Zj&index=2)

Video 3:

[https://www.youtube.com/watch?v=eGUMIqw1pzU&list=PLvL\\_GB22RnWJqRXbYcC5uIpHzupeqx6Zj&index=3](https://www.youtube.com/watch?v=eGUMIqw1pzU&list=PLvL_GB22RnWJqRXbYcC5uIpHzupeqx6Zj&index=3)

Video 4:

[https://www.youtube.com/watch?v=DkiIm\\_fwPmQ&list=PLvL\\_GB22RnWJqRXbYcC5uIpHzupeqx6Zj&index=4](https://www.youtube.com/watch?v=DkiIm_fwPmQ&list=PLvL_GB22RnWJqRXbYcC5uIpHzupeqx6Zj&index=4)

Video 5:

[https://www.youtube.com/watch?v=-f1bPBWgcTQ&list=PLvL\\_GB22RnWJqRXbYcC5uIpHzupeqx6Zj&index=5](https://www.youtube.com/watch?v=-f1bPBWgcTQ&list=PLvL_GB22RnWJqRXbYcC5uIpHzupeqx6Zj&index=5)

Video 6:

[https://www.youtube.com/watch?v=bCr0Wcs5LEY&list=PLvL\\_GB22RnWJqRXbYcC5uIpHzupeqx6Zj&index=6](https://www.youtube.com/watch?v=bCr0Wcs5LEY&list=PLvL_GB22RnWJqRXbYcC5uIpHzupeqx6Zj&index=6)

Video 7:

[https://www.youtube.com/watch?v=xCfdLxdw9Dc&list=PLvL\\_GB22RnWJqRXbYcC5uIpHzupeqx6Zj&index=7](https://www.youtube.com/watch?v=xCfdLxdw9Dc&list=PLvL_GB22RnWJqRXbYcC5uIpHzupeqx6Zj&index=7)

Video Bonus:

[https://www.youtube.com/watch?v=gY5nHXfWyEQ&list=PLvL\\_GB22RnWJqRXbYcC5uIpHzupeqx6Zj&index=8](https://www.youtube.com/watch?v=gY5nHXfWyEQ&list=PLvL_GB22RnWJqRXbYcC5uIpHzupeqx6Zj&index=8)

Buono Studio!!